QT       += core gui network

greaterThan(QT_MAJOR_VERSION, 4): QT += widgets

TEMPLATE = app

# You can make your code fail to compile if it uses deprecated APIs.
# In order to do so, uncomment the following line.
#DEFINES += QT_DISABLE_DEPRECATED_BEFORE=0x060000    # disables all the APIs deprecated before Qt 6.0.0

SOURCES += \
    global.cpp \
    main.cpp \
    qsingleapplication.cpp \
    settingsdialog.cpp \
    watermarkwindow.cpp \
    windowinfo.cpp \
    windowlistener.cpp \
    windowmanager.cpp

HEADERS += \
    global.h \
    macro.h \
    qsingleapplication.h \
    settingsdialog.h \
    watermarkwindow.h \
    windowinfo.h \
    windowlistener.h \
    windowmanager.h

FORMS += \
    settingsdialog.ui

RESOURCES += \
    resources.qrc

include ($$PWD/qhotkey/qhotkey.pri)
include ($$PWD/qrencode/qrencode.pri)
include ($$PWD/utils/utils.pri)

# ***********************************************************
# Win
# ***********************************************************
win32 {

    LIBS += -luser32

    CONFIG(debug, debug|release) {
        DESTDIR = $$PWD/../output/win/debug
    } else {
        DESTDIR = $$PWD/../output/win/release
    }
}

# ***********************************************************
# Linux
# ***********************************************************
linux {

    LIBS += -ldl

    HEADERS += \
        qsignalhandler.h

    SOURCES += \
        qsignalhandler.cpp

    CONFIG(debug, debug|release) {
        DESTDIR = $$PWD/../output/linux/debug
    } else {
        DESTDIR = $$PWD/../output/linux/release
    }
}
